

```

1  (* Comments in Standard ML begin with (* and end with *).  Comments can be
2     nested which means that all (* tags must end with a *) tag.  This comment,
3     for example, contains two nested comments. *)
4
5  (* A Standard ML program consists of declarations, e.g. value declarations: *)
6  val rent = 1200
7  val phone_no = 5551337
8  val pi = 3.14159
9  val negative_number = ~15  (* Yeah, unary minus uses the 'tilde' symbol *)
10
11 (* And just as importantly, functions: *)
12 fun is_large(x : int) = if x > 37 then true else false
13
14 (* Floating-point numbers are called "reals". *)
15 val tau = 2.0 * pi          (* You can multiply two reals *)
16 val twice_rent = 2 * rent    (* You can multiply two ints *)
17 (* val meh = 1.25 * 10 *)    (* But you can't multiply an int and a real *)
18
19 (* +, - and * are overloaded so they work for both int and real. *)
20 (* The same cannot be said for division which has separate operators: *)
21 val real_division = 14.0 / 4.0  (* gives 3.5 *)
22 val int_division = 14 div 4      (* gives 3, rounding down *)
23 val int_remainder = 14 mod 4     (* gives 2, since 3*4 = 12 *)
24
25 (* ~ is actually sometimes a function (e.g. when put in front of variables) *)
26 val negative_rent = ~(rent)     (* Would also have worked if rent were a "real" *)
27
28 (* There are also booleans and boolean operators *)
29 val got_milk = true
30 val got_bread = false
31 val has_breakfast = got_milk andalso got_bread  (* 'andalso' is the operator *)
32 val has_something = got_milk orelse got_bread    (* 'orelse' is the operator *)
33 val is_sad = not(has_something)                  (* not is a function *)
34
35 (* Many values can be compared using equality operators: = and <> *)
36 val pays_same_rent = (rent = 1300)  (* false *)
37 val is_wrong_phone_no = (phone_no <> 5551337)  (* false *)
38
39 (* The operator <> is what most other languages call !=. *)
40 (* 'andalso' and 'orelse' are called && and || in many other languages. *)
41
42 (* Actually, most of the parentheses above are unnecessary.  Here are some
43    different ways to say some of the things mentioned above: *)
44 fun is_large x = x > 37  (* The parens above were necessary because of ': int' *)
45 val is_sad = not has_something
46 val pays_same_rent = rent = 1300  (* Looks confusing, but works *)
47 val is_wrong_phone_no = phone_no <> 5551337
48 val negative_rent = ~rent  (* ~ rent (notice the space) would also work *)
49
50 (* Parentheses are mostly necessary when grouping things: *)
51 val some_answer = is_large (5 + 5)  (* Without parens, this would break! *)
52 (* val some_answer = is_large 5 + 5 *)  (* Read as: (is_large 5) + 5. Bad! *)
53
54
55 (* Besides booleans, ints and reals, Standard ML also has chars and strings: *)
56 val foo = "Hello, World!\n"  (* The \n is the escape sequence for linebreaks *)
57 val one_letter = #"a"        (* That funky syntax is just one character, a *)
58
59 val combined = "Hello " ^ "there, " ^ "fellow!\n"  (* Concatenate strings *)
60
61 val _ = print foo  (* You can print things. We are not interested in the *)
62 val _ = print combined  (* result of this computation, so we throw it away. *)
63 (* val _ = print one_letter *)  (* Only strings can be printed this way *)
64
65
66 val bar = [ #"H", #"e", #"l", #"l", #"o" ]  (* SML also has lists! *)
67 (* val _ = print bar *)  (* Lists are unfortunately not the same as strings *)
68
69 (* Fortunately they can be converted.  String is a library and implode and size

```

```

70  are functions available in that library that take strings as argument. *)
71  val bob = String.implode bar          (* gives "Hello" *)
72  val bob_char_count = String.size bob  (* gives 5 *)
73  val _ = print (bob ^ "\n")           (* For good measure, add a linebreak *)
74
75  (* You can have lists of any kind *)
76  val numbers = [1, 3, 3, 7, 229, 230, 248] (* : int list *)
77  val names = [ "Fred", "Jane", "Alice" ]  (* : string list *)
78
79  (* Even lists of lists of things *)
80  val groups = [ [ "Alice", "Bob" ],
81                [ "Huey", "Dewey", "Louie" ],
82                [ "Bonnie", "Clyde" ] ]    (* : string list list *)
83
84  val number_count = List.length numbers  (* gives 7 *)
85
86  (* You can put single values in front of lists of the same kind using
87     the :: operator, called "the cons operator" (known from Lisp). *)
88  val more_numbers = 13 :: numbers (* gives [13, 1, 3, 3, 7, ...] *)
89  val more_groups = ["Batman", "Superman"] :: groups
90
91  (* Lists of the same kind can be appended using the @ ("append") operator *)
92  val guest_list = [ "Mom", "Dad" ] @ [ "Aunt", "Uncle" ]
93
94  (* This could have been done with the "cons" operator. It is tricky because the
95     left-hand-side must be an element whereas the right-hand-side must be a list
96     of those elements. *)
97  val guest_list = "Mom" :: "Dad" :: [ "Aunt", "Uncle" ]
98  val guest_list = "Mom" :: ("Dad" :: ("Aunt" :: ("Uncle" :: [])))
99
100 (* If you have many lists of the same kind, you can concatenate them all *)
101 val everyone = List.concat groups (* [ "Alice", "Bob", "Huey", ... ] *)
102
103 (* A list can contain any (finite) number of values *)
104 val lots = [ 5, 5, 5, 6, 4, 5, 6, 5, 4, 5, 7, 3 ] (* still just an int list *)
105
106 (* Lists can only contain one kind of thing... *)
107 (* val bad_list = [ 1, "Hello", 3.14159 ] : ??? list *)
108
109
110 (* Tuples, on the other hand, can contain a fixed number of different things *)
111 val person1 = ("Simon", 28, 3.14159) (* : string * int * real *)
112
113 (* You can even have tuples inside lists and lists inside tuples *)
114 val likes = [ ("Alice", "ice cream"),
115              ("Bob", "hot dogs"),
116              ("Bob", "Alice") ]      (* : (string * string) list *)
117
118 val mixup = [ ("Alice", 39),
119              ("Bob", 37),
120              ("Eve", 41) ] (* : (string * int) list *)
121
122 val good_bad_stuff =
123   ([ "ice cream", "hot dogs", "chocolate" ],
124    [ "liver", "paying the rent" ])    (* : string list * string list *)
125
126
127 (* Records are tuples with named slots *)
128
129 val rgb = { r=0.23, g=0.56, b=0.91 } (* : {b:real, g:real, r:real} *)
130
131 (* You don't need to declare their slots ahead of time. Records with
132    different slot names are considered different types, even if their
133    slot value types match up. For instance... *)
134
135 val Hsl = { H=310.3, s=0.51, l=0.23 } (* : {H:real, l:real, s:real} *)
136 val Hsv = { H=310.3, s=0.51, v=0.23 } (* : {H:real, s:real, v:real} *)
137
138 (* ...trying to evaluate `Hsv = Hsl` or `rgb = Hsl` would give a type
139    error. While they're all three-slot records composed only of `real`s,

```

```

140   they each have different names for at least some slots. *)
141
142 (* You can use hash notation to get values out of tuples. *)
143
144 val H = #H Hsv (* : real *)
145 val s = #s Hsl (* : real *)
146
147 (* Functions! *)
148 fun add_them (a, b) = a + b    (* A simple function that adds two numbers *)
149 val test_it = add_them (3, 4) (* gives 7 *)
150
151 (* Larger functions are usually broken into several lines for readability *)
152 fun thermometer temp =
153   if temp < 37
154   then "Cold"
155   else if temp > 37
156        then "Warm"
157        else "Normal"
158
159 val test_thermo = thermometer 40 (* gives "Warm" *)
160
161 (* if-sentences are actually expressions and not statements/declarations.
162    A function body can only contain one expression. There are some tricks
163    for making a function do more than just one thing, though. *)
164
165 (* A function can call itself as part of its result (recursion!) *)
166 fun fibonacci n =
167   if n = 0 then 0 else (* Base case *)
168   if n = 1 then 1 else (* Base case *)
169   fibonacci (n - 1) + fibonacci (n - 2) (* Recursive case *)
170
171 (* Sometimes recursion is best understood by evaluating a function by hand:
172
173   fibonacci 4
174   ~> fibonacci (4 - 1) + fibonacci (4 - 2)
175   ~> fibonacci 3 + fibonacci 2
176   ~> (fibonacci (3 - 1) + fibonacci (3 - 2)) + fibonacci 2
177   ~> (fibonacci 2 + fibonacci 1) + fibonacci 2
178   ~> ((fibonacci (2 - 1) + fibonacci (2 - 2)) + fibonacci 1) + fibonacci 2
179   ~> ((fibonacci 1 + fibonacci 0) + fibonacci 1) + fibonacci 2
180   ~> ((1 + fibonacci 0) + fibonacci 1) + fibonacci 2
181   ~> ((1 + 0) + fibonacci 1) + fibonacci 2
182   ~> (1 + fibonacci 1) + fibonacci 2
183   ~> (1 + 1) + fibonacci 2
184   ~> 2 + fibonacci 2
185   ~> 2 + (fibonacci (2 - 1) + fibonacci (2 - 2))
186   ~> 2 + (fibonacci (2 - 1) + fibonacci (2 - 2))
187   ~> 2 + (fibonacci 1 + fibonacci 0)
188   ~> 2 + (1 + fibonacci 0)
189   ~> 2 + (1 + 0)
190   ~> 2 + 1
191   ~> 3  which is the 4th Fibonacci number, according to this definition
192
193 *)
194
195 (* A function cannot change the variables it can refer to. It can only
196    temporarily shadow them with new variables that have the same names. In this
197    sense, variables are really constants and only behave like variables when
198    dealing with recursion. For this reason, variables are also called value
199    bindings. An example of this: *)
200
201 val x = 42
202 fun answer(question) =
203   if question = "What is the meaning of life, the universe and everything?"
204   then x
205   else raise Fail "I'm an exception. Also, I don't know what the answer is."
206 val x = 43
207 val hmm = answer "What is the meaning of life, the universe and everything?"
208 (* Now, hmm has the value 42. This is because the function answer refers to
209    the copy of x that was visible before its own function definition. *)

```

```

210
211
212 (* Functions can take several arguments by taking one tuples as argument: *)
213 fun solve2 (a : real, b : real, c : real) =
214   ( (~b + Math.sqrt(b * b - 4.0*a*c)) / (2.0 * a),
215     (~b - Math.sqrt(b * b - 4.0*a*c)) / (2.0 * a) )
216
217 (* Sometimes, the same computation is carried out several times. It makes sense
218 to save and re-use the result the first time. We can use "let-bindings": *)
219 fun solve2 (a : real, b : real, c : real) =
220   let val discr = b * b - 4.0*a*c
221       val sqr = Math.sqrt discr
222       val denom = 2.0 * a
223   in ((~b + sqr) / denom,
224       (~b - sqr) / denom) end
225
226
227 (* Pattern matching is a funky part of functional programming. It is an
228 alternative to if-sentences. The fibonacci function can be rewritten: *)
229 fun fibonacci 0 = 0 (* Base case *)
230   | fibonacci 1 = 1 (* Base case *)
231   | fibonacci n = fibonacci (n - 1) + fibonacci (n - 2) (* Recursive case *)
232
233 (* Pattern matching is also possible on composite types like tuples, lists and
234 records. Writing "fun solve2 (a, b, c) = ..." is in fact a pattern match on
235 the one three-tuple solve2 takes as argument. Similarly, but less intuitively,
236 you can match on a list consisting of elements in it (from the beginning of
237 the list only). *)
238 fun first_elem (x::xs) = x
239 fun second_elem (x::y::xs) = y
240 fun evenly_positioned_elems (odd::even::xs) = even::evenly_positioned_elems xs
241   | evenly_positioned_elems [odd] = [] (* Base case: throw away *)
242   | evenly_positioned_elems [] = [] (* Base case *)
243
244 (* When matching on records, you must use their slot names, and you must bind
245 every slot in a record. The order of the slots doesn't matter though. *)
246
247 fun rgbToTup {r, g, b} = (r, g, b) (* fn : {b:'a, g:'b, r:'c} -> 'c * 'b * 'a *)
248 fun mixRgbToTup {g, b, r} = (r, g, b) (* fn : {b:'a, g:'b, r:'c} -> 'c * 'b * 'a *)
249
250 (* If called with {r=0.1, g=0.2, b=0.3}, either of the above functions
251 would return (0.1, 0.2, 0.3). But it would be a type error to call them
252 with {r=0.1, g=0.2, b=0.3, a=0.4} *)
253
254 (* Higher order functions: Functions can take other functions as arguments.
255 Functions are just other kinds of values, and functions don't need names
256 to exist. Functions without names are called "anonymous functions" or
257 lambda expressions or closures (since they also have a lexical scope). *)
258 val is_large = (fn x => x > 37)
259 val add_them = fn (a,b) => a + b
260 val thermometer =
261   fn temp => if temp < 37
262             then "Cold"
263             else if temp > 37
264                 then "Warm"
265                 else "Normal"
266
267 (* The following uses an anonymous function directly and gives "ColdWarm" *)
268 val some_result = (fn x => thermometer (x - 5) ^ thermometer (x + 5)) 37
269
270 (* Here is a higher-order function that works on lists (a list combinator) *)
271 val readings = [ 34, 39, 37, 38, 35, 36, 37, 37, 37 ] (* first an int list *)
272 val opinions = List.map thermometer readings (* gives [ "Cold", "Warm", ... ] *)
273
274 (* And here is another one for filtering lists *)
275 val warm_readings = List.filter is_large readings (* gives [39, 38] *)
276
277 (* You can create your own higher-order functions, too. Functions can also take
278 several arguments by "currying" them. Syntax-wise this means adding spaces
279 between function arguments instead of commas and surrounding parentheses. *)

```

```

280 fun map f [] = []
281   | map f (x::xs) = f(x) :: map f xs
282
283 (* map has type ('a -> 'b) -> 'a list -> 'b list and is called polymorphic. *)
284 (* 'a is called a type variable. *)
285
286
287 (* We can declare functions as infix *)
288 val plus = add_them (* plus is now equal to the same function as add_them *)
289 infix plus (* plus is now an infix operator *)
290 val seven = 2 plus 5 (* seven is now bound to 7 *)
291
292 (* Functions can also be made infix before they are declared *)
293 infix minus
294 fun x minus y = x - y (* It becomes a little hard to see what's the argument *)
295 val four = 8 minus 4 (* four is now bound to 4 *)
296
297 (* An infix function/operator can be made prefix with 'op' *)
298 val n = op + (5, 5) (* n is now 10 *)
299
300 (* 'op' is useful when combined with high order functions because they expect
301   functions and not operators as arguments. Most operators are really just
302   infix functions. *)
303 val sum_of_numbers = foldl op+ 0 [1,2,3,4,5]
304
305
306 (* Datatypes are useful for creating both simple and complex structures *)
307 datatype color = Red | Green | Blue
308
309 (* Here is a function that takes one of these as argument *)
310 fun say(col) =
311     if col = Red then "You are red!" else
312     if col = Green then "You are green!" else
313     if col = Blue then "You are blue!" else
314     raise Fail "Unknown color"
315
316 val _ = print (say(Red) ^ "\n")
317
318 (* Datatypes are very often used in combination with pattern matching *)
319 fun say Red = "You are red!"
320   | say Green = "You are green!"
321   | say Blue = "You are blue!"
322   | say _ = raise Fail "Unknown color"
323
324
325 (* Here is a binary tree datatype *)
326 datatype 'a btree = Leaf of 'a
327                   | Node of 'a btree * 'a * 'a btree (* three-arg constructor *)
328
329 (* Here is a binary tree *)
330 val myTree = Node (Leaf 9, 8, Node (Leaf 3, 5, Leaf 7))
331
332 (* Drawing it, it might look something like...
333
334       8
335      / \
336 leaf -> 9  5
337        / \
338 leaf -> 3  7 <- leaf
339 *)
340
341 (* This function counts the sum of all the elements in a tree *)
342 fun count (Leaf n) = n
343   | count (Node (leftTree, n, rightTree)) = count leftTree + n + count rightTree
344
345 val myTreeCount = count myTree (* myTreeCount is now bound to 32 *)
346
347
348 (* Exceptions! *)
349 (* Exceptions can be raised/thrown using the reserved word 'raise' *)

```



```

350 fun calculate_interest(n) = if n < 0.0
351     then raise Domain
352     else n * 1.04
353
354 (* Exceptions can be caught using "handle" *)
355 val balance = calculate_interest ~180.0
356     handle Domain => ~180.0    (* x now has the value ~180.0 *)
357
358 (* Some exceptions carry extra information with them *)
359 (* Here are some examples of built-in exceptions *)
360 fun failing_function []      = raise Empty  (* used for empty lists *)
361   | failing_function [x]    = raise Fail "This list is too short!"
362   | failing_function [x,y]  = raise Overflow (* used for arithmetic *)
363   | failing_function xs     = raise Fail "This list is too long!"
364
365 (* We can pattern match in 'handle' to make sure
366    a specific exception was raised, or grab the message *)
367 val err_msg = failing_function [1,2] handle Fail _ => "Fail was raised"
368                                           | Domain => "Domain was raised"
369                                           | Empty => "Empty was raised"
370                                           | _      => "Unknown exception"
371
372 (* err_msg now has the value "Unknown exception" because Overflow isn't
373    listed as one of the patterns -- thus, the catch-all pattern _ is used. *)
374
375 (* We can define our own exceptions like this *)
376 exception MyException
377 exception MyExceptionWithMessage of string
378 exception SyntaxError of string * (int * int)
379
380 (* File I/O! *)
381 (* Write a nice poem to a file *)
382 fun writePoem(filename) =
383     let val file = TextIO.openOut(filename)
384         val _ = TextIO.output(file, "Roses are red,\nViolets are blue.\n")
385         val _ = TextIO.output(file, "I have a gun.\nGet in the van.\n")
386     in TextIO.closeOut(file) end
387
388 (* Read a nice poem from a file into a list of strings *)
389 fun readPoem(filename) =
390     let val file = TextIO.openIn filename
391         val poem = TextIO.inputAll file
392         val _ = TextIO.closeIn file
393     in String.tokens (fn c => c = #"\n") poem
394     end
395
396 val _ = writePoem "roses.txt"
397 val test_poem = readPoem "roses.txt" (* gives [ "Roses are red,",
398                                           "Violets are blue.",
399                                           "I have a gun.",
400                                           "Get in the van." ] *)
401
402 (* We can create references to data which can be updated *)
403 val counter = ref 0 (* Produce a reference with the ref function *)
404
405 (* Assign to a reference with the assignment operator *)
406 fun set_five reference = reference := 5
407
408 (* Read a reference with the dereference operator *)
409 fun equals_five reference = !reference = 5
410
411 (* We can use while loops for when recursion is messy *)
412 fun decrement_to_zero r = if !r < 0
413     then r := 0
414     else while !r >= 0 do r := !r - 1
415
416 (* This returns the unit value (in practical terms, nothing, a 0-tuple) *)
417
418 (* To allow returning a value, we can use the semicolon to sequence evaluations *)
419 fun decrement_ret x y = (x := !x - 1; y)

```